

# COMPUTATIONS CONCERNING THE UNIFORM DOMINATION NUMBER OF A FINITE SIMPLE GROUP

TIMOTHY C. BURNES AND SCOTT HARPER

ABSTRACT. In this note we provide a record of the GAP and MAGMA computations used in the paper [6] on the uniform domination number of finite simple groups. The methods are similar to those used by Breuer, Guralnick and Kantor in [4] to study the uniform spread of simple groups, where the computations are documented in [3]. Throughout this note, we freely adopt the notation and terminology introduced in [6].

## 1. CODE

1.1. **GAP functions.** In this section we present two GAP [9] functions, which rely on the Character Table Library [2]. We will use the ATLAS notation [8], adopted by GAP, to refer to simple groups and their conjugacy classes. We will assume that the following packages are loaded:

```
LoadPackage( "AtlasRep" );
LoadPackage( "CTblLib" );
LoadPackage( "TomLib" );
```

1.1.1. *Maximal subgroups.* The function `MaximalOvergroups` provides information about the set  $\mathcal{M}(G, s)$  of maximal overgroups of an element  $s \in G$ . The input is the name of a group  $G$  and the name of a conjugacy class  $s^G$ . It returns a list of pairs  $[H, k]$  where  $H$  represents a conjugacy class of maximal subgroups of  $G$  and  $k$  is the number of conjugates of  $H$  which contain  $s$ .

```
MaximalOvergroups:= function( G, s )
  local tbl, mx, prim, spos, nonz, j, results;

  tbl:= CharacterTable( G );
  mx:= List( Maxes(tbl), CharacterTable );
  prim:= List( mx, x -> TrivialCharacter( x )^tbl );
  spos:= Position( AtlasClassNames( tbl ), s );
  nonz:= PositionsProperty( prim, pi -> pi[spos] <> 0 );

  results:= [];
  for j in nonz do
    Add(results, [ Identifier(mx[j]), prim[j][ spos ] ]);
  od;

  return results;
end;
```

For example, if  $G = \text{Suz}$  and  $s \in 14A$  then `MaximalOvergroups("Suz", "14A")` returns

```
[ [ J2.2, 2 ], [ (a4xps1(3,4)):2, 1 ] ]
```

and thus  $\mathcal{M}(G, s) = \{J_{2.2}, J_{2.2}, (A_4 \times L_3(4)):2\}$ .

We now define a related function `MU` which takes as input the name of a group  $G$  and returns a pair  $[\mu, L]$  where  $\mu = \mu(G) := \min_{s \in G} |\mathcal{M}(g, s)|$  and  $L$  is the list of names of conjugacy classes  $s^G$  with  $|\mathcal{M}(G, s)| = \mu(G)$ .

```
MU:= function( G )
  local tbl, mx, prims, sums, mu, mins;

  tbl:= CharacterTable( G );
  names:= AtlasClassNames( tbl );
  mx:= List( Maxes(tbl), CharacterTable );
  prims:= List( mx, x -> TrivialCharacter( x )^tbl );
  sums:=Sum(prims);
  mu:= Minimum(sums);
  mins:=List(PositionsProperty(sums,s -> s = mu), p-> names[p]);

  return [mu, mins];
end;
```

For example, `MU("Suz")` returns `[ 3, [ "14A", "21A", "21B" ] ]`. In particular, this shows that  $\mu(\text{Suz}) = 3$ .

1.1.2. *Probabilistic methods.* The function `ProbabilisticMethod` implements the probabilistic method described in [6, Section 2.2] to compute an upper bound on the uniform domination number. The function takes as input the name of a group  $G$  and the name of a conjugacy class  $s^G$ . It returns the least positive integer  $c$  such that  $\widehat{Q}(G, s, c) < 1$  (see [6, (2.3)]). If no such integer  $c$  exists, then the function will not halt.

```
ProbabilisticMethod:= function( G, s )
  local tbl, mx, prim, spos, nonz, fprs, tot, size, orde, leng, i, j,
  c, prob;

  tbl:= CharacterTable( G );
  mx:= List( Maxes(tbl), CharacterTable );
  prim:= List( mx, x -> TrivialCharacter( x )^tbl );
  spos:= Position( AtlasClassNames( tbl ), s );
  nonz:= PositionsProperty( prim, pi -> pi[spos] <> 0 );
  fprs:= List( prim, pi -> pi/pi[1] );
  size:= SizesConjugacyClasses( tbl );
  orde:= OrdersClassRepresentatives( tbl );
  leng:= Length( size );

  tot:=[];
  for i in [ 1 .. Length( nonz ) ] do
    tot:= tot + prim[ nonz[i] ][ spos ]*fprs[ nonz[i] ];
  od;

  c:=0;
  prob:=2;
  while prob > 1 do
    c:=c+1;
    prob:=0;
    for j in [ 2 .. leng ] do
      if IsPrime(orde[j]) then
```

```

        prob:= prob + size[j]*tot[j]^c;
    fi;
od;
od;
return c;
end;

```

For example, `ProbabilisticMethod("G2(3)","13A")` returns 5, which tells us that  $\gamma_u(G_2(3)) \leq 5$  (witnessed by the class 13A).

**1.2. MAGMA functions.** The character-theoretic methods described in Section 1.1 are not always available. For example, GAP may not store the character tables of all maximal subgroups of the group in question. In such cases we compute directly in the group, and we prefer to use MAGMA [1] for this. In this section we present our MAGMA functions.

**1.2.1. Fixed point ratios.** The function `FixedPointRatio` determines the fixed point ratio  $\text{fpr}(g, G/H)$  of an element  $g \in G$  in the action of  $G$  on the coset space  $G/H$ . It takes as input a group  $G$ , a subgroup  $H$  of  $G$  and an element  $g \in G$ . The function returns the fixed point ratio  $\text{fpr}(g, G/H)$ .

```

FixedPointRatio:= function( G, H, g )
    count:=0;
    classreps:=Classes(H);
    for rep in classreps do
        if (rep[1] eq Order(g)) then
            if (IsConjugate(G,g,rep[3])) then
                count:=count+rep[2];
            end if;
        end if;
    end for;
    return count*Order(Centraliser(G,g))/Order(G);
end function;

```

**1.2.2. Maximal subgroups.** The function `MaximalOvergroups` provides information about the maximal overgroups of an element. The input is a group  $G$  and an element  $s$  of  $G$ . The function returns a pair of lists  $[H_1, \dots, H_m]$  and  $[k_1, \dots, k_m]$ , where the  $H_i$  are pairwise non-conjugate maximal subgroups of  $G$  and  $k_i$  is the number of conjugates of  $H_i$  which contain  $s$ .

```

MaximalOvergroups:= function( G, s )
    groups:=[];
    mults:=[];
    maxes:=MaximalSubgroups(G : OrderMultipleOf:=Order(s));
    for M in maxes do
        H:=M'subgroup;
        count:=FixedPointRatio(G,H,s)*Order(G)/Order(H);
        if (count ne 0) then
            groups:=Append(groups,H);
            mults:=Append(mults,count);
        end if;
    end for;
    return <groups, mults>;
end function;

```

For example, suppose we wish to determine the maximal subgroups of the alternating group  $A_{11}$  which contain a fixed 11-cycle  $s$ . The following code

```

G:=AlternatingGroup(11);
s:=G!(1,2,3,4,5,6,7,8,9,10,11);
MaximalOvergroups(G,s);

returns

<[
  Permutation group acting on a set of cardinality 11
  Order = 7920 = 2^4 * 3^2 * 5 * 11
    (1, 8)(2, 10)(3, 11)(5, 7)
    (1, 11, 10, 9)(2, 4, 7, 6),
  Permutation group acting on a set of cardinality 11
  Order = 7920 = 2^4 * 3^2 * 5 * 11
    (1, 10)(2, 8)(3, 11)(5, 7)
    (1, 4, 7, 6)(2, 11, 10, 9)
], [ 1, 1 ]>

```

from which it is easy to verify that the maximal overgroups of  $s$  are two non-conjugate subgroups isomorphic to the Mathieu group  $M_{11}$  (runtime: 0.010s).

The corresponding function for computing  $\mu(G)$  is as follows:

```

MU:= function( G )
  cl:=Classes(G);
  mu:=&+MaximalOvergroups(G,cl[1][3])[2];
  for i in [2..#cl] do
    sum:=&+MaximalOvergroups(G,cl[i][3])[2];
    if sum lt mu then
      mu:=sum;
    end if;
  end for;
  return mu;
end function;

```

For example,  $MU(\text{AlternatingGroup}(11))$  returns 2, showing that  $\mu(A_{11}) = 2$  (runtime: 3.540s).

**1.2.3. Probabilistic methods.** Here we present a MAGMA implementation of the probabilistic method for bounding the uniform domination number. As before, the function takes as input a group  $G$  and an element  $s$  of  $G$ . It returns the least positive integer  $c$  such that  $\widehat{Q}(G, s, c) < 1$ . If no such integer  $c$  exists, then the function will not halt.

```

ProbabilisticMethod:= function( G, s )
  maxover:=MaximalOvergroups(G,s);
  max:=maxover[1];
  mult:=maxover[2];
  classes:=Classes(G);
  c:=0;
  prob:=2;
  while prob ge 1 do
    c:=c+1;
    prob:=0;
    for j in [ 2 .. #classes ] do
      if IsPrime(classes[j][1]) then
        tot:=0;
        for i in [1..#mult] do

```

```

        tot:=tot+mult[i]*FixedPointRatio(G,max[i],classes[j][3]);
    end for;
    prob:= prob + classes[j][2]*tot^c;
end if;
end for;
end while;
return c;
end function;

```

For example, consider the uniform domination number of the orthogonal group  $\Omega_{10}^+(2)$  with respect to a class of elements of order 51 (these elements act as  $2^- \perp 8^-$  on the natural module, see [6, Section 6.1]). Then

```

G:=POmegaPlus(10,2);
order:=0;
while order ne 51 do
    s:=Random(G);
    order:=Order(s);
end while;

```

```

ProbabilisticMethod(G,s);

```

returns 5, whence  $\gamma_u(\Omega_{10}^+(2)) \leq 5$  (runtime: 2.260s).

1.2.4. *Random and exhaustive searches.* In this section we present three examples to explain how we use MAGMA to conduct random and exhaustive searches for uniform dominating sets.

First we present a function `PrimeUnion`, which we will use in the examples below. This function takes as input a group  $G$  and an element  $s \in G$ , and it returns a set of elements which generate the prime order subgroups of the maximal subgroups of  $G$  containing  $s$ . Note that this function calls `MaximalOvergroups` (and hence `FixedPointRatio`).

```

PrimeUnion:= function( G, s )
    N:=Normaliser(G,sub<G|s>);
    V:=[];
    max:=MaximalOvergroups(G,s);
    for i in [1..#max[1]] do
        M:=max[1][i]; n:=max[2][i];
        A:=[];
        repeat
            repeat
                g:=Random(M);
                a,x:=IsConjugate(G,g,s);
            until a eq true;
            Mx:=Conjugate(M,x);
            orb:=Orbit(N,Set(Mx));
            for O in orb do
                if O in A eq false then
                    A:=Append(A, O);
                    for u in O do
                        if IsPrime(Order(u)) and
                            #({x : x in V} meet {u^i : i in [1..Order(u)]}) eq 0 then
                                V:=Include(V,u);
                            end if;
                        end for;
                    end for;
                end if;
            end for;
        end for;
    end for;
end function;

```

```

        end if;
    end for;
    until #A eq n;
end for;

    return V;
end function;

```

**Example.** To prove that  $A_{13}$  has a uniform dominating set consisting of two conjugate 13-cycles we proceed as follows. (We require `PrimeUnion`, `MaximalOvergroups` and `FixedPointRatio` to be defined.)

```

G:=Alt(13);
s:=G!(1,2,3,4,5,6,7,8,9,10,11,12,13);

V:=PrimeUnion(G,s);
done:=false;
for i in [1..1000] do
    x:=Random(G); x:=s^x;
    check:=true;
    for g in V do
        if not(#sub<G|x,g> eq #G) then
            check:=false;
            break;
        end if;
    end for;
    if check eq true then
        done:=true;
        print[s,x];
        break;
    end if;
end for;

```

This prints a uniform dominating set for  $A_{13}$  of size 2 (runtime: 3.560s).

**Example.** Similarly, to find four elements in  $10A$  which form a uniform dominating set for  $M_{12}$  we proceed as follows:

```

load m12;
c1:=Classes(G);
s:=c1[13][3];
V:=PrimeUnion(G,s);

done:=false;
for i in [1..1000] do
    x:=Random(G); x:=s^x;
    y:=Random(G); y:=s^y;
    z:=Random(G); z:=s^z;
    check:=true;
    for g in V do
        if not((#sub<G|x,g> eq #G) or (#sub<G|y,g> eq #G)
            or (#sub<G|z,g> eq #G)) then
            check:=false;
            break;
        end if;
    end for;
end for;

```

```

    end if;
end for;
if check eq true then
    done:=true;
    print[s,x,y,z];
    break;
end if;
end for;

```

This prints a uniform dominating set for  $M_{12}$  of size 4; see Proposition 2.4 (runtime: 1.310s).

**Example.** The following code can be used to prove that  $M_{12}$  does not have a uniform dominating set consisting of three elements in the class 6A:

```

load m12;
cl:=Classes(G);
s:=cl[9][3];
V:=PrimeUnion(G,s);
C:=Class(G,s);
U:={};

for c in C do
    if #({x : x in U} meet {c^i : i in [1..Order(c)]}) eq 0 then
        U:=Include(U,c);
    end if;
end for;

test:= function( x, y )
    success:=1;
    for g in V do
        if not((#sub<G|g,x> eq #G) or (#sub<G|g,y> eq #G)) then
            success:=0;
            break;
        end if;
    end for;
    return success;
end function;

U:=[x : x in U];
success:=0;
for i in [1..#U-1] do for j in [i+1..#U] do
    if test(U[i],U[j]) eq 1 then
        success:=1;
        break i;
    end if;
end for; end for;
print success;

```

This prints 0, which justifies the claim (runtime: 17613s; memory: 13 MB).

## 2. RESULTS

In this section we explain the computational results that are needed in [6], in terms of the GAP and MAGMA functions described in the previous section.

### 2.1. Alternating groups.

**Proposition 2.1.** *The values of  $\mu(A_n)$  in the following table are correct. Moreover, we have  $\mu(A_n) = |\mathcal{M}(A_n, s)|$  for any  $s \in A_n$  with the given cycle-shape.*

$n$	5	6	7	9	11	13	15	17	19	21	23
$\mu(A_n)$	1	2	2	3	2	3	3	2	1	3	2
$s$	[5]	[5, 1]	[7]	[5, 2 <sup>2</sup> ]	[11]	[9, 2 <sup>2</sup> ]	[11, 2 <sup>2</sup> ]	[17]	[19]	[17, 2 <sup>2</sup> ]	[23]

*Proof.* Use the functions `MU` and `MaximalOvergroups` in `GAP` if  $n \leq 13$ , and in `MAGMA` if  $n \geq 15$ .  $\square$

**Proposition 2.2.** *The information on the uniform domination number  $\gamma_u(A_n)$  in the following table is correct.*

$n$	5	6	7	9	11	13	15	17	23	31
$\gamma_u(A_n)$	3	4	4	$\leq 6$	3	2	$\leq 6$	2	2	2

*Proof.* We carry out all of these computations in `MAGMA`. Set  $G = A_n$ . For a proper subgroup  $H < G$ , let  $b(G, G/H)$  be the base size of the action of  $G$  on the coset space  $G/H$ .

First assume  $n = 5$ . To show that 3 is an upper bound we use `ProbabilisticMethod` with a 5-cycle. The lower bound is obtained by exhaustive searches on each conjugacy class. (This can also be verified by hand, using elementary arguments.)

Next suppose  $n = 6$ . By a random search we can find four conjugate 5-cycles which form a total dominating set for  $G$ , so  $\gamma_u(G) \leq 4$ . We show  $\gamma_u(G) \geq 4$  by exhaustive searches on each conjugacy class  $s^G$ . In fact, it is sufficient to assume that  $s$  has shape [4, 2] or [3, 3]; if not, then  $s$  is contained in the stabiliser  $H$  of a 1-set with  $b(G, G/H) = 4$  (see [6, Section 2.1] for further details).

The case  $n = 7$  is very similar. Here we carry out a random search with conjugate 7-cycles to show that  $\gamma_u(G) \leq 4$ , and we establish equality by means of an exhaustive search on each class  $s^G$  (by considering base sizes, it is helpful to note that we may assume  $s$  has shape [7] or [3, 2<sup>2</sup>]).

Now assume  $n = 11$ . To see that  $\gamma_u(G) \leq 3$ , use `ProbabilisticMethod` with an 11-cycle. To get equality, we apply the following results on base sizes. Let  $s \in G$ . If  $s$  is not an 11-cycle, then  $s$  is contained in an intransitive subgroup  $H$  and [11, Theorem 3.1] implies that  $b(G, G/H) \geq \lceil \log_2 n \rceil - 1 = 3$ . If  $s$  is an 11-cycle then  $s$  is contained in a subgroup  $H = M_{11}$  (see the proof of Proposition 2.1) and [5, Theorem 1] gives  $b(G, G/H) = 3$ . Therefore,  $\gamma_u(G) \geq 3$  and thus  $\gamma_u(G) = 3$  as claimed.

For  $n = 13$ , a random search identifies two conjugate 13-cycles which form a uniform dominating set (see the first example in Section 1.2.4). Indeed, one can check that  $\{s_1, s_2\}$  has the desired property, where

$$\begin{aligned} s_1 &= (1, 2, 3, 4, 5, 6, 7, 8, 9, 10, 11, 12, 13) \\ s_2 &= (1, 2, 3, 4, 5, 6, 8, 9, 12, 7, 11, 10, 13). \end{aligned}$$

Finally, if  $n \in \{9, 15, 17, 23, 31\}$  then use `ProbabilisticMethod` with an  $n$ -cycle.  $\square$

### 2.2. Sporadic simple groups.

**Proposition 2.3.** *For each sporadic simple group  $G$ , the value of  $\mu(G)$  is given in the second column of Table 1. Moreover,  $\mu(G) = |\mathcal{M}(G, s)|$  for  $s \in G$  in the conjugacy class recorded in the fourth column.*

*Proof.* If  $G = \mathbb{B}$  or  $\mathbb{M}$ , then [10, Table IV] gives  $\mu(G) = |\mathcal{M}(G, s)| = 1$ . In the remaining cases, use `MU` and `MaximalOvergroups` in `GAP`.  $\square$



$G$	$\mu(G)$	$\gamma_u(G)$	$s$	$\mathcal{M}(G, g)$	$b$	$c$
$M_{11}$	1	4	11A	$L_2(11)$	4	
$M_{12}$	3	4	10A	$A_6.2^2, A_6.2^2, 2 \times S_5$		6
$M_{22}$	1	3	11A	$L_2(11)$	3	
$M_{23}$	1	2	23A	23:11	2	
$M_{24}$	2	3 or 4	21A	$L_3(4):S_3, 2^6:(L_3(2) \times S_3)$		4
$J_1$	1	2	15A	$D_6 \times D_{10}$	2	
$J_2$	3	3 or 4	10C	$2^{1+4}:A_5, A_5 \times D_{10}, 5^2:D_{12}$		4
$J_3$	2	2 or 3	19A	$L_2(19), L_2(19)$		3
$J_4$	1	2	43A	43:14	2	
HS	2	3 or 4	15A	$S_8, 5:4 \times A_5$		4
Suz	3	3	14A	$J_2:2, J_2:2, (A_4 \times L_3(4)):2$		3
McL	3	3	15A	$3^{1+4}:2.S_5, 2.A_8, 5^{1+2}:3:8$		3
Ru	1	2	29A	$L_2(29)$	2	
He	1	2 or 3	17A	$Sp_4(4):2$	4	
			21A	$3.S_7, 7^{1+2}:(3 \times S_3),$ $7:3 \times L_3(2), 7:3 \times L_3(2)$		3
Ly	1	2	28A	$2.A_{11}$	2	
O'N	2	2	31A	$L_2(31), L_2(31)$		2
Co <sub>1</sub>	1	2 or 3	26A	$(A_4 \times G_2(4)):2$	3	
Co <sub>2</sub>	1	3	23A	$M_{23}$	3	
Co <sub>3</sub>	1	3	23A	$M_{23}$	3	
Fi <sub>22</sub>	1	3 or 4	22A	$2.U_6(2)$	5	
			16A	$2^{5+8}:(S_3 \times A_6), 2.2^{1+8}:(U_4(2):2)$ $2^{10}:M_{22}, {}^2F_4(2)' (4 \text{ times})$	4	
Fi <sub>23</sub>	1	2	35A	$S_{12}$	2	
Fi' <sub>24</sub>	1	2	29A	29:14	2	
HN	1	2 or 3	22A	2.HS.2	3	
Th	2	2	19A	$U_3(8):6, L_2(19):2$		2
$\mathbb{B}$	1	2	47A	47:23	2	
$\mathbb{M}$	1	2	59A	$L_2(59)$	2	

TABLE 1. Sporadic simple groups

**Proposition 2.4.** *Let  $G$  be a sporadic simple group. Then*

$$d - \varepsilon \leq \gamma_u(G) \leq d,$$

where  $d$  is defined as follows:

$G$	$M_{11}$	$M_{12}$	$M_{22}$	$M_{23}$	$M_{24}$	$J_1$	$J_2$	$J_3$	$J_4$	HS	Suz	McL	Ru
$d$	4	4	3	2	4*	2	4*	3*	2	4*	3	3	2
	He	Ly	O'N	Co <sub>1</sub>	Co <sub>2</sub>	Co <sub>3</sub>	Fi <sub>22</sub>	Fi <sub>23</sub>	Fi' <sub>24</sub>	HN	Th	$\mathbb{B}$	$\mathbb{M}$
	3*	2	2	3*	3	3	4*	2	2	3*	2	2	2

Here an asterisk indicates that  $\varepsilon = 1$ ; otherwise  $\varepsilon = 0$  and  $\gamma_u(G) = d$ . In particular,  $\gamma_u(G) \leq 4$ , with equality if  $G = M_{11}$  or  $M_{12}$ .

*Proof.* If  $\mu(G) = 1$ , then choose  $s \in G$  such that  $\mathcal{M}(G, s) = \{H\}$  and the base size  $b(G, G/H)$  is minimal (the base size of every almost simple primitive group with sporadic socle has been computed; see [7, 12]), and note that  $\gamma_u(G) \leq b(G, G/H)$  (see [6, Section 2.1]). The minimal base size is denoted by  $b$  in Table 1. In particular,  $\gamma_u(G) = 2$  if  $G = \mathbb{B}$  or  $\mathbb{M}$ .

For each sporadic simple group  $G \notin \{\mathbb{B}, \mathbb{M}\}$  and each class representative  $s \in G$  we use the function `ProbabilisticMethod` in `GAP` to obtain an upper bound  $\gamma_u(G) \leq c$ . For the groups with  $\mu(G) = 1$ , we almost always find that  $b \leq c$ ; the exceptions are the cases  $G \in \{\text{He}, \text{Fi}_{22}\}$ , where it is better to apply the probabilistic method with an element  $s \in G$  with  $|\mathcal{M}(G, s)| > 1$ . For these two groups, and also for those with  $\mu(G) > 1$ , we record the minimal value of  $c$  in Table 1, together with an element  $s \in G$  such that  $\widehat{Q}(G, s, c) < 1$ . For example, if  $G = \text{He}$  and  $s \in 17\text{A}$  then Table 1 indicates that  $\mathcal{M}(G, s) = \{H\}$  with  $H = \text{Sp}_4(4).2$  and  $b(G, G/H) = 4$ . However, if we choose  $s \in 21\text{A}$  then  $|\mathcal{M}(G, s)| = 4$  and  $\widehat{Q}(G, s, 3) < 1$ , so  $\gamma_u(G) \in \{2, 3\}$ .

To derive a lower bound on  $\gamma_u(G)$ , we again exploit the connection with base sizes. For example, if  $G = \text{M}_{11}$  then every element of  $G$  is contained in a subgroup  $H$  isomorphic to  $\text{L}_2(11)$  or  $\text{M}_{10}$ ; in both cases  $b(G, G/H) = 4$ , which implies that  $\gamma_u(G) \geq 4$ . With the exception of  $G = \text{M}_{12}$ , this explains how we obtain the results on  $\gamma_u(G)$  presented in Table 1.

For  $G = \text{M}_{12}$ , this approach only gives  $3 \leq \gamma_u(G) \leq 6$ , but by carrying out a random search in `MAGMA` (see Section 1.2.4) one can show that the class `10A` contains a uniform dominating set of size 4. For example, if

$$G = \langle (4, 10)(6, 8)(7, 9)(11, 12), (1, 10, 2)(3, 7, 11)(4, 6, 5)(8, 12, 9) \rangle \cong \text{M}_{12}$$

then  $\{s_1, s_2, s_3, s_4\}$  has the desired property, where

$$\begin{aligned} s_1 &= (1, 6, 11, 5, 3, 2, 7, 9, 4, 8)(10, 12) \\ s_2 &= (1, 2, 7, 5, 8, 12, 11, 10, 9, 4)(3, 6) \\ s_3 &= (1, 8)(2, 11, 9, 7, 12, 4, 5, 3, 10, 6) \\ s_4 &= (1, 9, 12, 10, 2, 6, 11, 5, 7, 3)(4, 8) \end{aligned}$$

This allows us to conclude that  $\gamma_u(G) \in \{3, 4\}$ .

In fact, we claim that  $\gamma_u(G) = 4$ . To see this, we proceed as follows. First, we use the `GAP` function `MaximalOvergroups` to determine  $\mathcal{M}(G, s)$  for an element  $s$  in each non-trivial conjugacy class of  $G$ . By combining the base size results in [7] with [6, Corollary 2.3], we quickly reduce the problem to the conjugacy classes labelled `3B` and `6A`. In each case, we need to show that the given conjugacy class does not contain a uniform dominating set of size 3. In fact, since the square of an element in `6A` is in the class `3B`, we only need to consider the class `6A`. The required exhaustive search can now be carried out in `MAGMA`, using the code presented in the final example of Section 1.2.4.  $\square$

### 2.3. Exceptional groups.

**Proposition 2.5.** *The values of  $\mu(G)$  in the following table are correct. Moreover, we have  $\mu(G) = |\mathcal{M}(G, s)|$  for  $s \in G$  in the given conjugacy class.*

$G$	$F_4(2)$	${}^2F_4(2)'$	$G_2(3)$	$G_2(4)$
$\mu(G)$	2	2	3	1
$s$	17A	16A	13A	21A

*Proof.* For  $G \neq F_4(2)$ , we use `MU` and `MaximalOvergroups` in `GAP`, so for the remainder let us assume  $G = F_4(2)$ . In this case, the function `Maxes` in `GAP` does not return the maximal subgroups of  $G$ ; however, `GAP` does contain the character table, and associated fusion maps, of the maximal subgroups of  $G$  isomorphic to  $\text{Sp}_8(2)$  and  $(2^{1+8} \times 2^6) : \text{Sp}_6(2)$ . We manually adapt our previous code from `ProbabilisticMethod` as follows:

```
tbl:= CharacterTable( "F4(2)" );
names:=AtlasClassNames( tbl );
NamesOfFusionSources( tbl );
```

```
partial:=["(2^(1+8)x2^6):S6(2)", "F4(2)M2", "S8(2)", "F4(2)M4"];
mx:= List(partial, CharacterTable );
sums:= Sum(List( mx, x -> TrivialCharacter( x )^tbl ));
p0:=List(PositionsProperty(sums,s -> s = 0), p-> names[p]);
p1:=List(PositionsProperty(sums,s -> s = 1), p-> names[p]);
```

This verifies that every element of  $G$  is contained in at least two maximal subgroups of this type, except for elements in 13A, which are contained in none, as well as elements in 21A, 21B, 28A, 28B, which are in exactly one.

The subgroups  $L_4(3)$ ,  ${}^3D_4(2).3$  and  ${}^2F_4(2)$  all contain elements of order 13, and since 13A is the unique class in  $G$  of elements of order 13, any 13A element must be contained in at least one conjugate of each of these subgroups.

The subgroup  ${}^3D_4(2)$  contains elements of order 21 and 28. Since there is a unique class of elements of order 21 in  $\text{Aut}(F_4(2))$ , the maximal overgroups of elements in 21A and 21B will be isomorphic. The same applies to elements of order 28. In particular, every element of order 21 and 28 is contained in a subgroup isomorphic to  ${}^3D_4(2)$ .

Therefore, every element of  $G$  is contained in at least two maximal subgroups. By [10, Proposition 6.2], if  $s$  has order 17 then  $\mathcal{M}(G, s) = \{H, K\}$  where  $H \cong K \cong \text{Sp}_8(2)$ . Therefore,  $\mu(F_4(2)) = 2$  and the proof of the proposition is complete.  $\square$

**Proposition 2.6.** *The upper bounds on  $\gamma_u(G)$  in the following table are correct.*

$G$	$F_4(2)$	${}^2F_4(2)'$	$G_2(3)$
$\gamma_u(G) \leq$	5	4	5
$s$	17A	16A	13A

*Proof.* For  $G = F_4(2)$  we proceed as in the proof of Proposition 2.5:

```
tbl:= CharacterTable( "F4(2)" );
partial:=["S8(2)", "F4(2)M4"];
mx:= List(partial, CharacterTable );
prim:= List( mx, x -> TrivialCharacter( x )^tbl );
spos:= Position( AtlasClassNames( tbl ), "17A" );
fprs:= List( prim, pi -> pi/pi[1] );
size:= SizesConjugacyClasses( tbl );
orde:= OrdersClassRepresentatives( tbl );
leng:= Length( size );

tot:= prim[1][ spos ]*fprs[1] + prim[2][ spos ]*fprs[2];

c:= 0;
prob:= 2;
while prob > 1 do
  c:= c+1;
  prob:=0;
  for j in [ 2 .. leng ] do
    if IsPrime(orde[j]) then
      prob:= prob + size[j]*tot[j]^c;
    fi;
  od;
od;

Print(c);
```

In the two remaining cases, we use `ProbabilisticMethod` in `GAP`, with an element  $s \in G$  as indicated in the table.  $\square$

#### 2.4. Classical groups.

**Proposition 2.7.** *The values of  $\mu(G)$  in the following table are correct. Moreover, we have  $\mu(G) = |\mathcal{M}(G, s)|$  for the given element  $s \in G$ .*

$G$	$\mathrm{P}\Omega_8^+(3)$	$\Omega_8^+(2)$	$\Omega_7(3)$	$\mathrm{U}_6(2)$	$\mathrm{Sp}_6(2)$	$\mathrm{U}_4(3)$
$\mu(G)$	7	7	3	4	2	5
$s$	14A	15A	14A	11A	15A	9A

*Proof.* In all cases, use `MU` and `MaximalOvergroups` in `GAP`.  $\square$

**Proposition 2.8.** *The upper bounds on  $\gamma_u(G)$  in the following table are correct.*

$G$	$\Omega_{10}^+(2)$	$\mathrm{L}_9(2)$	$\mathrm{L}_8(2)$	$\mathrm{P}\Omega_8^+(3)$	$\mathrm{L}_7(2)$	$\mathrm{PSp}_6(3)$
$\gamma_u(G) \leq$	5	7	9	19	6	3
$s$	$2^- \perp 8^-$	$4 \oplus 5$	$3 \oplus 5$	14A	105A	14A

  

$\mathrm{U}_6(2)$	$\mathrm{L}_4(2)$	$\mathrm{U}_4(3)$	$\mathrm{U}_4(2)$	$\mathrm{U}_3(5)$	$\mathrm{L}_2(7)$
6	4	16	8	13	3
11A	15A	9A	9A	13A	7A

*Proof.* In each case we use the function `ProbabilisticMethod`, with the element given in the table. More precisely, we use `MAGMA` when  $G = \Omega_{10}^+(2)$ ,  $\mathrm{L}_9(2)$  or  $\mathrm{L}_8(2)$  (see the example in Section 1.2.3 for the case  $G = \Omega_{10}^+(2)$ ), and `GAP` in the remaining cases.  $\square$

### 3. RESOURCES

For the computations, we use a combination of `GAP` Version 4.5.6 and `MAGMA` 2.19-2, on a 2.7GHz machine with 128 GB RAM. The character-theoretic computations run quickly in `GAP` and we adopt this approach whenever possible. The computations in `MAGMA` for determining maximal overgroups of specific elements and implementing the probabilistic approach (via fixed point ratios) are more resource-intensive, but still feasible for the groups we are interested in. For example, an implementation of the probabilistic method applied to  $\mathrm{L}_9(2)$  with an element of order 465 (see Proposition 2.8) can be done in 616 seconds, using 771 MB of memory. Similar resources are needed for most of the exhaustive searches in `MAGMA`, which we use to rule out the existence of total domination sets with prescribed properties. As discussed in Section 2.2, the verification of the bound  $\gamma_u(\mathrm{M}_{12}) \geq 4$  is a notable exception in terms of runtime. For example, the computation to rule out the existence of a uniform domination set of size 3 in the class `6A` was timed at 17613 seconds (using 13 MB of memory).

### REFERENCES

- [1] W. Bosma, J. Cannon and C. Playoust, *The MAGMA algebra system I: The user language*, J. Symb. Comput. **24** (1997), 235–265.
- [2] T. Breuer, *The GAP Character Table Library, Version 1.2.1*, `GAP` package, <http://www.math.rwth-aachen.de/~Thomas.Breuer/ctbllib>, 2012.
- [3] T. Breuer, *GAP computations concerning probabilistic generation of finite simple groups*, preprint (arxiv:0710.3267).
- [4] T. Breuer, R.M. Guralnick and W.M. Kantor, *Probabilistic generation of finite simple groups, II*, J. Algebra **320** (2008), 443–494.
- [5] T.C. Burness, R.M. Guralnick and J. Saxl, *On base sizes for symmetric groups*, Bull. London Math. Soc. **44** (2011), 386–391.
- [6] T.C. Burness and S. Harper, *On the uniform domination number of a finite simple group*, preprint (arxiv:1710.07113).

- [7] T.C. Burness, E.A. O'Brien and R.A. Wilson, *Base sizes for sporadic groups*, Israel J. Math. **177** (2010), 307–333.
- [8] J.H. Conway, R.T. Curtis, S.P. Norton, R.A. Parker and R.A. Wilson, *Atlas of Finite Groups*, Oxford University Press, 1985.
- [9] The GAP Group, *GAP – Groups, Algorithms, and Programming*, Version 4.8.7, 2017, (<http://www.gap-system.org>).
- [10] R.M. Guralnick and W.M. Kantor, *Probabilistic generation of finite simple groups*, J. Algebra **234** (2000), 743–792.
- [11] Z. Halasi, *On the base size of the symmetric group acting on subsets*, Stud. Sci. Math. Hung. **49** (2012), 492–500.
- [12] M. Neunhöffer, F. Noeske, E.A. O'Brien and R.A. Wilson, *Orbit invariants and an application to the Baby Monster*, J. Algebra **341** (2011), 297–305.

T.C. BURNES, SCHOOL OF MATHEMATICS, UNIVERSITY OF BRISTOL, BRISTOL BS8 1TW, UK  
*E-mail address:* `t.burness@bristol.ac.uk`

S. HARPER, SCHOOL OF MATHEMATICS, UNIVERSITY OF BRISTOL, BRISTOL BS8 1TW, UK  
*E-mail address:* `scott.harper@bristol.ac.uk`